

Web Science & Engineering
Semantic Web Practical Assignment

Andrea Alfieri, 5128315

October 4, 2019

1. Open Questions

- **What is an “ontology” in computer science?**

While philosophy defines *ontology* as the branch of metaphysics dealing with the nature of being, computer science considers an ontology as a set of concepts and the relationship between those concepts, which only exist within a domain. It can be used to define a domain and to reason about the properties of that domain. It is basically the way computer scientists try to describe everything that *exists* in the real world (which, again, brings us back to the philosophical problem of defining what *is*), but also their property and relationships.

- **What is a “knowledge base”?**

A *knowledge base* is a technology used to store and access complex information by a computer system. For a greater precision and completeness, different knowledge bases may focus on different domains to be then linked for a more comprehensive view.

- **What are the core differences and similarities between SPARQL and SQL?**

SQL and SPARQL are both query languages, but the first one is used to access information from a database, while the second one is used to retrieve and query information from a RDF graph.

SPARQL is also capable of extracting subgraphs or even creating new RDF graphs based on information in the queried graphs.

- **Why are “knowledge graph embeddings” useful?**

KGE are very useful for building recommendations by clustering and similarity measurements, and for completing and building knowledge bases by guessing relationships and entity types, or by extracting and discovering new entities and relationships.

2.2 All museums in the Netherlands

```
#defaultView:Map
SELECT ?coords
WHERE {
  #?museum is an instance of any subclass of museum
  ?museum wdt:P31/wdt:P279* wd:Q33506.

  #?museum has country Netherlands
  ?museum wdt:P17 wd:Q55.

  #?museum has coords ?coords
  ?museum wdt:P625 ?coords
}
```




Figure 1: All museums in the Delft & Den Haag area

2.3 Safe Money on Tickets

```
#defaultView:ImageGrid
SELECT ?image ?paintingLabel ?artistLabel
WHERE {
  #?painting instance of painting
  ?painting wdt:P31 wd:Q3305213.

  #?painting has location Rijksmuseum Amsterdam
  ?painting wdt:P276 wd:Q190804.

  #?painting has image ?image and has creator ?artist
  ?painting wdt:P18 ?image;
            wdt:P170 ?artist.

  SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}
ORDER BY ?artistLabel
```




Figure 2: All the paintings in Rijksmuseum Amsterdam

2.4 Who is a busy painter?

As show in Figure 3, Jean Baptiste Vanmour is the most productive artist in the Rijksmuseum.
The museum owns 27 Rembrandt.

```

SELECT ?artistLabel (COUNT(?painting) AS ?numberOfPaintings)
WHERE {
    #?painting instance of painting
    ?painting wdt:P31 wd:Q3305213.

    #?painting has location Rijksmuseum Amsterdam
    ?painting wdt:P276 wd:Q190804.

    #?painting has creator ?artist
    ?painting wdt:P170 ?artist.

    SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}
GROUP BY ?artistLabel
ORDER BY DESC(?numberOfPaintings)

```

artistLabel	numberOfPaintings
anonymous	707
Jean Baptiste Vanmour	61
George Hendrik Breitner	56
Jacob Maris	54
Arnoud van Halen	47
Jan Maurits Quinkhard	38
Michiel van Mierevelt	37
Pieter van der Werff	37
Jozef Israëls	37
Johannes Gijsbert Vogel	36
Jan van Ravesteyn	32
Caspar Netscher	28
Anton Mauve	28
Rembrandt	27
Johannes Bosboom	26
Ferdinand Bol	25
Cornelis Troost	25
Jan Steen	23
Gerard de Lairesse	23
Jurriaen Andriessen	23

Figure 3: All the most productive artists in Rijksmuseum

2.5.1 Paintings and countries

To derive the country where every painting is located, I simply used the *location* property and then used the *country* property on the extracted location. Many paintings had multiple values for the location, but most of those were inside the same country, so I easily fixed those entries just by adding the DISTINCT statement at the beginning of the query. The problem with “A Young Woman Seated at the Virginals” is that its locations differ through different countries, so my strategy has been to try extract the *end time* from each location, to then be able to use GROUP BY ?paintingLabel and (MAX(?endTime) AS ?maxTime) to retrieve the most recent location of the painting (Russia). I successfully managed to extract the end date with the following statements:

```
?painting p:P276 ?statement.    # "?painting": "location": ?statement
?statement ps:P276 ?location.    # value
?statement pq:P582 ?endTime.    # qualifier
```

but I could not complete the “cleaning” of the results because many paintings use slightly different standards, so many entries disappeared from the final table. This is the most precise result I was able to obtain, which has 43 lines:

```
SELECT DISTINCT ?paintingLabel ?countryLabel
WHERE {
    #?painting instance of painting
    ?painting wdt:P31 wd:Q3305213.

    #?painting has creator Vermeer
    ?painting wdt:P170 wd:Q41264.

    #?painting has location which has country ?country
    ?painting wdt:P276 [ wdt:P17 ?country ].

    SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}
ORDER BY ?paintingLabel
```

Painting Name	Country
A Girl Asleep	United States of America
A Lady Writing a Letter	United States of America
A Young Woman Seated at the Virginals	Russia
A Young Woman Seated at the Virginals	France
A Young Woman Seated at the Virginals	United States of America
A Young Woman Seated at the Virginals	United Kingdom
A Young Woman Seated at the Virginals	Japan
A Young Woman Seated at the Virginals	People's Republic of China
A Young Woman Seated at the Virginals	Italy
Christ in the House of Martha and Mary	United Kingdom
Diana and Her Companions	Netherlands
Girl Interrupted at her Music	United States of America
Girl Reading a Letter at an Open Window	Germany
Girl with a Flute	United States of America
Girl with a Pearl Earring	Netherlands
Girl with a Red Hat	United States of America
Lady Seated at a Virginal	United Kingdom
Lady Standing at a Virginal	United Kingdom
Lady Writing a Letter with her Maid	Ireland
Mistress and Maid	United States of America
Officer and Laughing Girl	United States of America
Portrait of a Young Woman	United States of America
Saint Praxedis	Japan
The Allegory of Faith	United States of America
The Art of Painting	Austria
The Astronomer	France
The Concert	United States of America
The Geographer	Germany
The Girl with the Wine Glass	Germany
The Guitar Player	United Kingdom
The Lacemaker	France
The Little Street	Netherlands
The Love Letter	Netherlands
The Milkmaid	Netherlands
The Music Lesson	United Kingdom
The Procuress	Germany
The Wine Glass	Germany
View of Delft	Netherlands
Woman Holding a Balance	United States of America
Woman in Blue Reading a Letter	Netherlands
Woman with a Lute	United States of America
Woman with a Pearl Necklace	Germany
Woman with a Water Jug	United States of America

2.5.2 Paintings and Cities

This time, I simply added the `wdt:P131` property (the item is located on the territory of the following administrative entity). The results went from 43 to 50 because the `DISTINCT` statement is now considering three columns instead of two, which shows that some paintings changed cities inside the same country over time.

```
SELECT DISTINCT ?paintingLabel ?countryLabel ?cityLabel
WHERE {
  #?painting instance of painting
  ?painting wdt:P31 wd:Q3305213.

  #?painting has creator ?artist
  ?painting wdt:P170 wd:Q41264.

  #?painting has location which has country ?country
  ?painting wdt:P276 [ wdt:P17 ?country;
                      wdt:P131 ?city ]. # NEW PART

  SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}
ORDER BY ?paintingLabel
```


Painting	Country	City
A Girl Asleep	United States of America	Manhattan
A Young Woman Seated at the Virginals	Russia	Khamovniki District
A Young Woman Seated at the Virginals	France	Saint-Germain-l'Auxerrois
A Young Woman Seated at the Virginals	Russia	Saint Petersburg
A Young Woman Seated at the Virginals	United States of America	Manhattan
A Young Woman Seated at the Virginals	United Kingdom	Central London
A Young Woman Seated at the Virginals	United States of America	Philadelphia
A Young Woman Seated at the Virginals	United Kingdom	Oxford
A Young Woman Seated at the Virginals	United States of America	Dallas
A Young Woman Seated at the Virginals	Japan	Ueno
A Young Woman Seated at the Virginals	United States of America	Alabama
A Young Woman Seated at the Virginals	People's Republic of China	Beijing
A Young Woman Seated at the Virginals	United Kingdom	Cambridge
A Young Woman Seated at the Virginals	United States of America	Raleigh
A Young Woman Seated at the Virginals	United States of America	Norfolk
A Young Woman Seated at the Virginals	Italy	Rome
Christ in the House of Martha and Mary	United Kingdom	City of Edinburgh
Christ in the House of Martha and Mary	United Kingdom	Edinburgh
Diana and Her Companions	Netherlands	The Hague
Girl Interrupted at her Music	United States of America	Manhattan
Girl Reading a Letter at an Open Window	Germany	Innere Altstadt
Girl with a Pearl Earring	Netherlands	The Hague
Lady Seated at a Virginal	United Kingdom	Central London
Lady Standing at a Virginal	United Kingdom	Central London
Lady Writing a Letter with her Maid	Ireland	Dublin
Mistress and Maid	United States of America	Manhattan
Officer and Laughing Girl	United States of America	Manhattan
Portrait of a Young Woman	United States of America	Manhattan
Saint Praxedis	Japan	Taito-ku
Saint Praxedis	Japan	Ueno Imperial Grant Park
The Allegory of Faith	United States of America	Manhattan
The Art of Painting	Austria	Innere Stadt
The Astronomer	France	1st arrondissement of Paris
The Concert	United States of America	Boston
The Geographer	Germany	Frankfurt
The Girl with the Wine Glass	Germany	Brunswick
The Guitar Player	United Kingdom	London Borough of Camden
The Lacemaker	France	1st arrondissement of Paris
The Little Street	Netherlands	Amsterdam
The Love Letter	Netherlands	Amsterdam
The Milkmaid	Netherlands	Amsterdam
The Music Lesson	United Kingdom	City of Westminster
The Procuress	Germany	Innere Altstadt
The Wine Glass	Germany	Berlin-Mitte
View of Delft	Netherlands	The Hague
Woman Holding a Balance	United States of America	Pennsylvania
Woman in Blue Reading a Letter	Netherlands	Amsterdam
Woman with a Lute	United States of America	Manhattan
Woman with a Pearl Necklace	Germany	Berlin-Mitte
Woman with a Water Jug	United States of America	Manhattan